#
# SPDX-License-Identifier: BSD-3-Clause
#
# Copyright © 2021 Keith Packard
#
# Redistribution and use in source and binary forms, with or without
# modification, are permitted provided that the following conditions
# are met:
#
# 1. Redistributions of source code must retain the above copyright
#    notice, this list of conditions and the following disclaimer.
#
# 2. Redistributions in binary form must reproduce the above
#    copyright notice, this list of conditions and the following
#    disclaimer in the documentation and/or other materials provided
#    with the distribution.
#
# 3. Neither the name of the copyright holder nor the names of its
#    contributors may be used to endorse or promote products derived
#    from this software without specific prior written permission.
#
# THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
# "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
# LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS
# FOR A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE
# COPYRIGHT HOLDER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT,
# INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
# (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR
# SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION)
# HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT,
# STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
# ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED
# OF THE POSSIBILITY OF SUCH DAMAGE.
#

if cc.get_define('_CALL_ELF') == '1'

  lib_semihost_srcs = [
    'opal_call.S',
    'opal_console_write.c',
    'opal_cec_power_down.c',
    'powerpc_exit.c',
    'powerpc_kill.c',
    'powerpc_stub.c',
  ]

  if tinystdio
    lib_semihost_srcs += 'powerpc_io.c'
  endif

  has_semihost = true

  foreach params : targets
    target = params['name']
    target_dir = params['dir']
    target_c_args = params['c_args']
    target_lib_prefix = params['lib_prefix']

    instdir = join_paths(lib_dir, target_dir)

    test_flags = []
    skipping = false
    foreach tf : target_c_args
      if skipping
	skipping = false
      elif tf == '-include'
	skipping = true
      else
	test_flags += tf
      endif
    endforeach

    if cc.get_define('__PPC64__', args : test_flags) != ''
      libsemihost_name = 'semihost'

      local_lib_semihost_target = static_library(join_paths(target_dir, target_lib_prefix + libsemihost_name),
						 lib_semihost_srcs,
						 install : really_install,
						 install_dir : instdir,
						 include_directories : inc,
						 c_args : target_c_args + c_args,
						 pic: false)

      set_variable('lib_semihost' + target, local_lib_semihost_target)
    endif

  endforeach

endif
